PharmaSUG China 2024 - Paper SA-10001

An Easy Way of Obtaining One-sided p-value for Asymptotic Generalized Cochran-Mantel-Haenszel (CMH) Test Between Proportions

Shaoming Liu, Fang Wang, Pfizer Inc

ABSTRACT

Cochran-Mantel-Haenszel (CMH) method is widely used when comparing two treatments in a stratified trial with a binary endpoint, using a weighted averaging of the stratum-specific differences between proportions. Within context of drug development for testing hypothesis, either p-values for CMH test statistic of general association or p-values from multiply imputed data with CMH test statistics are more commonly applied. When alternative hypothesis is one-sided, there is no SAS procedure (*SAS*®) or options in SAS procedure to directly output one-sided p-value for asymptotic generalized CMH test for differences between proportions. A cursory search of the methods of computing one-sided p-value for asymptotic generalized CMH test has been conducted but not many articles of the same topic had turned up, but we might have missed it.

This paper provides a quick guidance to obtain one-sided p-value for asymptotic generalized CMH test from SAS PROC FREQ statement.

Keywords: Asymptotic generalized CMH test, One-sided p-value

INTRODUCTION

This article introduces the derivation of one-sided p-value for asymptotic generalized CMH test when the primary endpoint is a proportion difference between two treatment groups and alternative hypothesis is one-sided. The computation of either CMH asymptotic generalized point estimate or corresponding confidence interval are not within the scope of this paper.

NOTATIONS AND FORMULAS

As we will see below, the CMH test compares binary responses of two treatment groups, adjusting for stratification factors. In the CMH test, the data are arranged in a series of associated 2 × 2 contingency tables, the null hypothesis is that the observed response is independent of the treatment used in any 2 × 2 contingency table.

Let O_{hij} be the observed frequency of treatment i (i = 1 or 2) with outcome j (j =1 or 2) in stratum h and E_{hij} be the expected frequency of treatment i (i = 1 or 2) with outcome j (j =1 or 2) in stratum h (the same notations are used for the rest of the paper), where

$$E_{hij} = \frac{(O_{hi1} + O_{hi2})(O_{h1j} + O_{h2j})}{O_{h11} + O_{h12} + O_{h21} + O_{h22}}$$

Also, let P_{hij} be the probability of outcome being j given treatment being i in stratum h. To test the following hypothesis,

$$H_0 : P_{h11} = P_{h21} \; for \; all \; h \in \{1, \ldots, H\}$$

versus

$$H_1 : P_{h11} > P_{h21} \; for \; at \; least \; one \; h \in \{1, \ldots, H\}$$

The asymptotic generalized CMH Test Statistics for large sample sizes is constructed as:

$$Z_{CMH} = \frac{\sum_{h=1}^{H} (O_{h11} - E_{h11})}{\sqrt{V_{11}}} \sim Z$$

Where

$$V_{11} = Var(\sum_{h=1}^{H}(O_{h11} - E_{h11}))$$

$$= \sum_{h=1}^{H} \frac{(O_{h11} + O_{h12})(O_{h21} + O_{h22})(O_{h11} + O_{h21})(O_{h12} + O_{h22})}{n^2(n-1)} = \sum_{h=1}^{H} \frac{E_{h11} \cdot E_{h22}}{n-1}$$

$$n = O_{h11} + O_{h12} + O_{h21} + O_{h22}$$

The test statistic Z_{CMH} will be compared with standard normal distribution to obtain p-value of the asymptotic generalized CMH test. From above description, the steps of calculating one-sided p-value of asymptotic generalized CMH test is to: 1) calculate Z_{CMH} 2) decide rejection region by alternative hypothesis and obtain one-sided p-value from normal distribution given Z_{CMH}.

STEPS OF CALCULATING Z_{CMH}

The CMH Test Statistics χ^2_{CMH} for testing general association between treatment and outcome of interest being i = 1 and j = 1 is

$$\chi^2_{CMH} = \frac{\left(\sum_{h=1}^{H}(O_{h11} - E_{h11})\right)^2}{\sum_{h=1}^{H} Var(O_{h11})}$$

Where for stratum h:

$$E_{h11} = \frac{(O_{h11} + O_{h12})(O_{h11} + O_{h21})}{O_{h11} + O_{h12} + O_{h21} + O_{h22}}$$

$$Var(O_{h11}) = \frac{(O_{h11} + O_{h12})(O_{h21} + O_{h22})(O_{h11} + O_{h21})(O_{h12} + O_{h22})}{n^2(n-1)}$$

For large samples, χ^2_{CMH} follows a χ^2distribution asymptotically with 1 df under the null hypothesis that there is no association between the treatment and the outcome. For stratified 2x2 tables, it's easy to deduce from above formulas that Z_{CMH} is the same with the squared root of χ^2_{CMH} except for the fact that Chi-square is always positive (+) due to its squared nature and thus sign of Z_{CMH} is to be determined. It is straightforward that the mathematical sign of the numerator of Z_{CMH}, a.k.a. $\sum_{h=1}^{H}(O_{h11} - E_{h11})$ determines sign of Z_{CMH}, which means Z_{CMH} equals to $\sqrt{\chi^2_{CMH}}$ if result of $\sum_{h=1}^{H} O_{h11} - \sum_{h=1}^{H} E_{h11}$ is positive value, and Z_{CMH} equals to $-\sqrt{\chi^2_{CMH}}$ if result of $\sum_{h=1}^{H} O_{h11} - \sum_{h=1}^{H} E_{h11}$ is negative value.

It is easy to obtain χ^2_{CMH} for testing general association via SAS PROC FREQ procedure with CMH option.

WORKING EXAMPLES WITH SAS IMPLEMENTATION

For example, Table 1 is a 2 x 2 contingency table of example dataset with four stratums, subjects are randomized to either active drug or placebo and have either clinical response or non-response.

		Response	Non-Response
Stratum 1	Active	12	3
	Placebo	2	5
Stratum 2	Active	9	10
	Placebo	0	9

Stratum 3	Active	46	31
	Placebo	10	28
Stratum 4	Active	23	22
	Placebo	5	17

Table 1. 2 x 2 Contingency Table of Example Dataset

To input the data into SAS datasets, one can use the following SAS data step statement:

```
data mydata;
       length strata $ 8 Treatment $ 8 Response $ 20;
       input strata $ Treatment $ Response $ Count @@;
       datalines;
1 Active Response 12 1 Active Non-response 3
1 Placebo Response 2 1 Placebo Non-response 5
2 Active Response 9 2 Active Non-response 10
2 Placebo Response 0 2 Placebo Non-response 9
3 Active Response 46 3 Active Non-response 31
3 Placebo Response 10 3 Placebo Non-response 28
4 Active Response 23 4 Active Non-response 22
4 Placebo Response 5 4 Placebo Non-response 17
;
```

It's easy to get χ^2_{CMH} with SAS:

```
proc freq data=mydata;
   tables strata*Treatment*Response /cmh expected norow nocol nopercent;
   weight Count;
   ods output cmh=chisq;
run;
```

χ^2_{CMH} from above SAS procedure for testing general association is 25.8460 (see Output 1):

Frequency Expected

Table 1 of Treatment by Response
Controlling for strata=1

Treatment	Response: Non-resp	Response: Response	Total
Active	3 5.4545	12 9.5455	15
Placebo	5 2.5455	2 4.4545	7
Total	8	14	22

Frequency Expected

Table 2 of Treatment by Response
Controlling for strata=2

Treatment	Response: Non-resp	Response: Response	Total
Active	10 12.893	9 6.1071	19
Placebo	9 6.1071	0 2.8929	9
Total	19	9	28

Frequency Expected

Table 3 of Treatment by Response
Controlling for strata=3

Treatment	Response: Non-resp	Response: Response	Total
Active	31 39.504	46 37.496	77
Placebo	28 19.496	10 18.504	38
Total	59	56	115

Frequency Expected

Table 4 of Treatment by Response
Controlling for strata=4

Treatment	Response: Non-resp	Response: Response	Total
Active	22 26.194	23 18.806	45
Placebo	17 12.806	5 9.194	22
Total	39	28	67

Cochran-Mantel-Haenszel Statistics (Based on Table Scores)				
Statistic	Alternative Hypothesis	DF	Value	Prob
1	Nonzero Correlation	1	25.8460	<.0001
2	Row Mean Scores Differ	1	25.8460	<.0001
3	General Association	1	25.8460	<.0001

Output 1. Output from PROC FREQ Statement

Also from 2 x 2 contingency table (Table 1), we can easily get O_{h11} and E_{h11} $h \in \{1, ..., 4\}$.

	Stratum 1	Stratum 2	Stratum 3	Stratum 4
$\boldsymbol{O_{h11}}$	12	9	46	23
$\boldsymbol{E_{h11}}$	9.5455	6.1071	37.496	18.805

The mathematical sign of Z_{CMH} is assessed to be positive (+):

$$\sum_{h=1}^{4}(O_{h11} - E_{h11}) = \sum_{h=1}^{4} O_{h11} - \sum_{h=1}^{4} E_{h11} = 90 - 71.9536 > 0$$

Therefore, $Z_{CMH} = (+)\sqrt{\chi^2_{CMH}} = 5.0839$.

For alternative hypothesis $H_1 : P_{h11} > P_{h21}\ for\ at\ least\ one\ h \in \{1, ..., 4\}$, the upper bound p-value is obtained by performing SAS function 1-probnorm(5.0839) = 1.8488133E-7;

For alternative hypothesis $H_1 : P_{h11} < P_{h21}\ for\ at\ least\ one\ h \in \{1, ..., 4\}$, the lower bound p-value is obtained by performing SAS function probnorm(5.0839) = 0.9999998151.

MACRO IMPLEMENTATION FOR ONE-SIDED CMH TEST AND ITS P-VALUE

```
%macro one_sided_cmh_test(indata=mydata
       ,treatment_of_interest_value=Active
       ,response_of_interest_value=Response
       ,side=upper /* test side of alternative hypothesis, value is upper or lower depending
on study hypothesis design defined in SAP */
       );

%* obtain Chi-squared CMH test statistic *;
proc freq data=mydata;
   tables strata*Treatment*Response /cmh expected norow nocol nopercent;
   weight Count;
   ods output cmh=chisq CrossTabFreqs=ctf;
run;

%* determine mathmatical sign of Zcmh *;
%*** filter out records of Treatment of Interest and Response of Interest by user input values
***;
data ctf1;
       set ctf;
       if _TYPE_="111" AND Treatment="&treatment_of_interest_value" AND
Response="&response_of_interest_value" then output;
run;
```

```
proc sql noprint;
	select sum(frequency) into: Observed from ctf1;
	select sum(expected) into: Expected from ctf1;
quit;

%* test statistic Zcmh and one-sided P-value *;
data Statistics;
	set chisq;
	where althypothesis="General Association";
	%if %sysevalf(&Observed-&Expected)>=0 %then
		Zcmh=sqrt(value);
	%else
		Zcmh=(-1)*sqrt(value);
	;
	%if %lowcase(&side)=upper %then
		one_sided_Pvalue=1-probnorm(Zcmh);
	%else %if %lowcase(&side)=lower %then
		one_sided_Pvalue=probnorm(Zcmh);
	;
	format Pvalue;
	rename value=chisq;
	label value="Cochran-Mantel-Haenszel Statistic (Chi-Square Statistic)" Zcmh="asymptotic
generalized CMH Test Statistic"  one_sided_Pvalue="one-sided p-value of asymptotic generalized
CMH test";
	keep value Zcmh one_sided_Pvalue;
run;

proc print data=Statistics label noobs;
title "One-sided p-value for Asymptotic Generalized Cochran-Mantel-Haenszel (CMH) Test";
run;

%mend;

%one_sided_cmh_test;
```

DISCUSSION AND CONCLUSION

Besides the Z_{CMH} approach introduced above, we can also obtain one-sided p-value by using p-value of χ^2_{CMH}. The null hypothesis for the χ^2_{CMH} assumes conditional odds ratio within each stratum equals to 1, which mirrors that of the Z_{CMH}, asserting that the conditional odds ratio equals 1 when proportions are equal and deviates from 1 when proportions differ. When conditional odds ratio within each stratum is equal to 1, the CMH statistic should be close to zero, while some or all conditional odds ratios is greater than 1, or some or all conditional odds ratios is less than 1, the CMH statistic is large. If we want to obtain one-sided p-value directly from p-value of χ^2_{CMH}, the sign of $\sum_{h=1}^{H}(O_{h11}-E_{h11})$ should be determined likewise. For example, when we test alternative hypothesis that the conditional odds ratio in some stratum is greater than 1, we can halve p-value of χ^2_{CMH} if $\sum_{h=1}^{H}(O_{h11}-E_{h11})$ is positive; Nevertheless, we should use 1-$\frac{1}{2}$ p-value of χ^2_{CMH} if $\sum_{h=1}^{H}(O_{h11}-E_{h11})$ is negative.

Take above χ^2_{CMH} of 25.8460 as an example. The p-value of χ^2_{CMH} is 3.6976472E-7 and $\sum_{h=1}^{H}(O_{h11}-E_{h11})$ is positive, halving the p-value we obtain 1.8488236E-7, which equals to the one-sided p-value from proposed Z_{CMH} approach when alternative hypothesis is $H_1: P_{h11} > P_{h21}\ for\ at\ least\ one\ h \in \{1, \ldots, H\}$.

Likewise , when alternative hypothesis is $H_1: P_{h11} < P_{h21}\ for\ at\ least\ one\ h \in \{1, \ldots, H\}$, the lower bound p-value is calculated using 1-$\frac{1}{2}$ p-value of χ^2_{CMH}.

Obs	Table	Statistic	AltHypothesis	DF	Value	Prob
1	Summary for Treatment * Response	1	Nonzero Correlation	1	25.8460	3.6976472E-7
2	Summary for Treatment * Response	2	Row Mean Scores Differ	1	25.8460	3.6976472E-7
3	Summary for Treatment * Response	3	General Association	1	25.8460	3.6976472E-7

It is worth to note, when heterogeneity across strata is present in data, in other words some conditional odds ratios are less than 1 and others are greater than 1, neither χ^2_{CMH} nor Z_{CMH} approach is as effective. Both approaches rely on the assumption of homogeneity across strata, and when this assumption is violated, an alternative approach becomes necessary.

In summary, the paper describes easy SAS implementation to compute one-sided p-value for asymptotic generalized CMH test. The results can be easily calculated. The R package *coin* provides the same asymptotic generalized CMH test with one-sided p-value output. We verified the steps we introduced provide equivalent results for the example in the literature.

REFERENCES

Agresti, A. (2002). *Categorical Data Analysis*, Second Edition. Hoboken, New Jersey: John Wiley & Sons.

Armitage, P. (1955). Tests for linear trends in proportions and frequencies. *Biometrics* **11**(3), 375–386. doi: 10.2307/3001775

Cochran, W.G. (1954). Some methods for strengthening the common *χ^2* tests. *Biometrics* **10**(4), 417–451. doi: 10.2307/3001616

Davis, L. J. (1986). Exact tests for *2 x 2* contingency tables. *The American Statistician* **40**(2), 139–141. doi: 10.1080/00031305.1986.10475377

Landis, J. R., Heyman, E. R. and Koch, G. G. (1978). Average partial association in three-way contingency tables: a review and discussion of alternative tests. *International Statistical Review* **46**(3), 237–254. doi: 10.2307/1402373

Mantel, N. and Haenszel, W. (1959). Statistical aspects of the analysis of data from retrospective studies of disease. *Journal of the National Cancer Institute* **22**(4), 719–748. doi: 10.1093/jnci/22.4.719

Mantel, N. (1963). Chi-square tests with one degree of freedom: extensions of the Mantel-Haenszel procedure. *Journal of the American Statistical Association* **58**(303), 690–700. doi: 10.1080/01621459.1963.10500879

Yates, F. (1948). The analysis of contingency tables with groupings based on quantitative characters. *Biometrika* **35**(1/2), 176–181. doi: 10.1093/biomet/35.1-2.176.

STAT 504 Analysis of Discrete Data, accessed 01 June 2024, <https://online.stat.psu.edu/stat504/lesson/5/5.3/5.3.5 >.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Shaoming Liu Shaoming.Liu@pfizer.com	Fang Wang Fang.Wang5@pfier.com